


SPSS Examples

Mediation

Background: Mediation modeling is a powerful analytical tool that can be used to explain the nature of the relationship among three or more variables. In addition to showing this simple relationship, it can be used to show how a variable mediates the relationship between levels of intervention and outcome. Recent works provide a statistical test for the mediation path known as the "Sobel Test." While



interactive web-based and stand alone methods exist for computing the Sobel Test, it is useful to have a program that will run the required regression analyses and compute the test statistic automatically.

You can download the data files for the SAS example as a Winzip file by clicking on program files.

Syntax

```
*****
* This program estimates the percentage of the total effect that is mediated *
* and the ratio of the indirect to the direct effect by using the Sobel test *
* Written by William Dudley PhD and Jose Benuzillo MA *
* University of Utah College of Nursing *
* 12/13/2002 *
*****.
```

```
DEFINE !data_in()
'a:\mediation\spss\example Data\'
!ENDDEFINE.
```

```
DEFINE !dat_out()
'a:\mediation\spss\regression files\'
!ENDDEFINE.
```

```
GET
FILE=!data_in + 'Example.sav'.
exe.
```

*define your own path, this is just a suggested one.

```
compute iv = Pain . /*your IV*/
compute dv = Depress. /*your DV*/
compute mediator = Function. /*you MEDIATOR*/
exe.
```

```
*****
* Regression analysis with the IV predicting the MEDIATOR *
*****.
```

```
REGRESSION
/MISSING LISTWISE
/STATISTICS COEFF OUTS R ANOVA
/CRITERIA=PIN(.05) POUT(.10)
/NOORIGIN
/DEPENDENT mediator
/METHOD=ENTER iv
/OUTFILE=COVB(!dat_out + 'reg1.sav') .
```

*define your own path, this is just
a suggested one.

```
*****  
* Regression analysis with IV and MEDIATOR predicting DV *  
*****.
```

```
REGRESSION  
/MISSING LISTWISE  
/STATISTICS COEFF OUTS R ANOVA  
/CRITERIA=PIN(.05) POUT(.10)  
/NOORIGIN  
/DEPENDENT dv  
/METHOD=ENTER iv mediator  
/OUTFILE=COVB(!dat_out + 'reg2.sav').
```

*define your own path, this is just
a suggested one.

```
*****  
* Regression analysis with IV predicting DV *  
*****.
```

```
REGRESSION  
/MISSING LISTWISE  
/STATISTICS COEFF OUTS R ANOVA  
/CRITERIA=PIN(.05) POUT(.10)  
/NOORIGIN  
/DEPENDENT dv  
/METHOD=ENTER iv  
/OUTFILE=COVB(!dat_out + 'reg3.sav').
```

*define your own path, this is just
a suggested one.

```
*****  
* Here we compute a variable named reg to identify the three different *  
* regression analysis we have run *  
*****.
```

```
get file=!dat_out + 'reg1.sav'.  
compute reg=1.  
exe.  
SAVE OUTFILE=!dat_out + 'reg1.sav'  
/COMPRESSED.
```

```
get file = !dat_out + 'reg2.sav'.  
compute reg=2.  
exe.  
SAVE OUTFILE=!dat_out + 'reg2.sav'  
/COMPRESSED.
```

```
get file=!dat_out + 'reg3.sav'.  
compute reg=3.  
exe.  
SAVE OUTFILE=!dat_out + 'reg3.sav'  
/COMPRESSED.
```

```
*****  
* Here we concatenate our three files reg1 reg2 and reg3 *  
*****.
```

```
add files file=!dat_out + 'reg1.sav'
         /file=!dat_out + 'reg2.sav'
         /file=!dat_out + 'reg3.sav'
/keep=  reg depvar_ rowtype_ varname_  const_ iv mediator.
exe.
```

```
*****
* Here we select only the statistics required to compute the Sobel          *
* equation (unstandardized coefficients and standard errors)                *
*****.
```

```
sel if rowtype_ = 'EST' or rowtype_ = 'SE' or rowtype_ = 'DFE'
      or rowtype_ = 'SIG'.
exe.
```

```
*****
* In this step we reshape our wide data file into long form                *
*****.
```

```
numeric a.
numeric sa.
numeric siga.
numeric b.
numeric sb.
numeric sigb.
numeric df.
numeric t.
numeric st.
numeric sig.
if reg = 1 and rowtype_ = 'EST' a    = iv.
if reg = 1 and rowtype_ = 'SE' sa   = iv.
if reg = 1 and rowtype_ = 'SIG' siga = iv.
if reg = 2 and rowtype_ = 'EST' b    = mediator.
if reg = 2 and rowtype_ = 'SE' sb   = mediator.
if reg = 2 and rowtype_ = 'SIG' sigb = mediator.
if reg = 2 and rowtype_ = 'DFE' df   = mediator.
if reg = 2 and rowtype_ = 'EST' tprime = iv.
if reg = 3 and rowtype_ = 'EST' t    = iv.
if reg = 3 and rowtype_ = 'SE' st   = iv.
if reg = 3 and rowtype_ = 'SIG' sig  = iv.
```

```
exe.
```

```
compute regx=1.
exe.
```

```
aggregate outfile !dat_out + 'sobel.sav'
  /break regx
  /a = max(a)
  /sa=max(sa)
  /sig= max(siga)
  /b=max(b)
  /sb=max(sb)
  /sigb=max(sigb)
  /df=max(df)
  /sig=max(sig)
  /tprime = max(tprime)
  /t=max(t)
```

```

/st=max(st).
get file !dat_out + 'sobel.sav'.

*define your own path, this is just
a suggested one.

format a b tprime t (F8.4).

compute ab = a*b.
compute ttprime = t-tprime.

format ab ttprime (F8.4).

exe.

*****
* Here we compute the sobel test and calculate the percentage of the total*
* effect that is mediated and the ratio of the indirect to the direct    *
* effect                                                                    *
* Further information about these tests, may be found in MacKinnon & Dwyer*
*(1993) Estimating mediated effects in prevention studies                 *
*****.

compute sobel= ttprime/(sqrt (( (b*b)*(sa*sa) ) + ( (a*a)*(sb*sb) ))).
compute absobel= abs(sobel).
compute p_val=2*(1-cdfnorm(absobel)).
compute t1=(t-(a*b)).
compute toteff=(a*b/((a*b)+t1)).
compute ratio=((a*b)/t1).
compute toteff = 100* toteff.
compute goodman = ttprime/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))+((sa*sa)*(sb*sb))).
compute absgood = abs(goodman).
compute goodman2 = ttprime/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))-((sa*sa)*(sb*sb))).
compute absgood2 = abs(goodman2).
compute p_val2 = 2*(1-cdfnorm(absgood)).
compute p_val3 = 2*(1-cdfnorm(absgood2)).
exe.

format p_val p_val2 p_val3
      sig siga sigb
      sobel goodman goodman2
      toteff ratio st sb sa(F8.6).
exe.

variable label
sig      'P value of c'
siga     'P value of a'
sigb     'P value of b'
a        'Reg coeff for the association between IV and MEDIATOR'
sa       'Standard error of a'
b        'Reg coeff for the association between the MEDIATOR and IV on DV'
sb       'Standard error of b'
df       'Degrees of freedom'
t        'Reg coeff for the association between IV and DV'
st       'Standard error of c'
sobel    'Sobel'
p_val    'P value'
goodman  'Goodman test'
p_val2   'P value'
goodman2 'GoodmanII test'
p_val3   'P Value'
toteff  'Percentage of the total effect that is mediated'
ratio    'Ratio of the indirect to the direct effect'.
exe.

```

* Basic Tables.

TEMPORARY.

TABLES

/FORMAT BLANK MISSING('')

/OBSERVATION t st sig a sa siga b sb sigb

/TABLES (t + st + sig + a + sa + siga + b + sb + sigb)

BY (STATISTICS)

/TITLE 'Regression Analysis Results'.

* Basic Tables.

TEMPORARY.

TABLES

/FORMAT BLANK MISSING('')

/OBSERVATION sobel p_val toteff ratio goodman p_val2 goodman2 p_val3

/TABLES (sobel + p_val + toteff + ratio + goodman + p_val2 + goodman2 + p_val3)

BY (STATISTICS)

/TITLE 'Mediation Analysis Results'.

* Basic Tables.

TEMPORARY.

TABLES

/FORMAT BLANK MISSING('')

/OBSERVATION toteff ratio

/TABLES (toteff + ratio)

BY (STATISTICS)

/TITLE 'Percent Mediated'.